

Rich export SPARQL queries: Award

Note that these are CONSTRUCT queries designed to create a small Jena model for export as a whole after a series of queries has been run. The PERSON_URI variable is substituted by VIVO at runtime.

award.sparql

```
PREFIX core: <http://vivoweb.org/ontology/core#>

CONSTRUCT {
    ?award ?property ?object .
} WHERE {
    PERSON_URI core:relatedBy ?awardReceipt .
    ?awardReceipt a core:AwardReceipt .
    ?awardReceipt core:relates ?award .
    ?award a core:Award .
    ?award ?property ?object .
}
```

conferringOrganization.sparql

```
PREFIX core: <http://vivoweb.org/ontology/core#>
PREFIX foaf: <http://xmlns.com/foaf/0.1/>

CONSTRUCT {
    ?organization ?property ?object .
} WHERE {
    PERSON_URI core:relatedBy ?awardReceipt .
    ?awardReceipt a core:AwardReceipt .
    ?awardReceipt core:assignedBy ?organization .
    ?organization a foaf:Organization .
    ?organization ?property ?object .
}
```

sponsoringOrganization.sparql

```
PREFIX core: <http://vivoweb.org/ontology/core#>
PREFIX foaf: <http://xmlns.com/foaf/0.1/>

CONSTRUCT {
    ?organization ?property ?object .
} WHERE {
    PERSON_URI core:relatedBy ?awardReceipt .
    ?awardReceipt a core:AwardReceipt .
    ?awardReceipt core:relates ?award .
    ?award a core:Award .
    ?award core:sponsoredBy ?organization .
    ?organization ?property ?object .
}
```